
Van: Kevin Erskine <KErskine@So ware-By-Ragazzi.com>

Here is what I use - You will need StringTheory if you want to use Asis

Original from: Fix BULLET issue with CW RTF 05/18/18 : Joseph Pena
<pena.joseph.michael@gmail.com>

I added a few tweaks...

You are free to use and modify as you like.

Kevin Erskine

RTF::equ_CRLF            EQUATE('<13,10>')
RTF::equ_Par             EQUATE(' \par' & RTF::equ_CRLF)
RTF::equ_Bar             EQUATE('|')

Cornerstone_RTF_Class.FixDocument  PROCEDURE (<STRING prmDocument>) !,STRING,PROC
MyString1       StringTheory

MyDoc       StringTheory
MyFontChar  StringTheory
MyLine      StringTheory
MyWord      StringTheory
wrkDlmBgn   LONG
wrkDlmEnd   LONG
wrkDlmPos   LONG
wrkLineNo   LONG
wrkPosBgn   LONG
wrkPosEnd   LONG
wrkReplace &STRING
wrkTrace    BYTE(0)
wrkWrdCnt   LONG

CODE
!----------------------------------------------------
!- Fix BULLET issue with CW RTF 05/18/18 : Joseph Pena <pena.joseph.michael@gmail.com>
!- RTF wouldn't be so bad if SoftVelocity would support the most recent versions RTF (version
2.0 instead of 4.1).
!- https://msdn.microsoft.com/en-us/library/windows/desktop/bb787873(v=vs.85).aspx
!----------------------------------------------------
IF OMITTED(prmDocument)
   MyDoc.SetValue(Self.Document.GetValue(),True)
ELSE
   MyDoc.SetValue(prmDocument,True)
END

MyDoc.Split(RTF::equ_CRLF)

DO HandleFixDoc

MyDoc.Join(RTF::equ_CRLF)

IF OMITTED(prmDocument)
   Self.DocSet(MyDoc.GetValue())
END

RETURN MyDoc.GetValue()
!-
!==============================================================================================

HandleFixDoc             ROUTINE
!-
  LOOP wrkLineNo = 1 to MyDoc.Records()
     MyLine.SetValue(MyDoc.GetLine(wrkLineNo))

     !implementing parsing for lists. 5/31/2018 - JMP
     IF MyLine.Instring('\listlevel')
        CYCLE
     END

     IF MyLine.Instring('\listtext')
        !replace unicode character with ansi
        MyLine.Replace('\u10625?', '\<39>95')

        !get rid of list link to override table symbol
        MyLine.Split('\')
        LOOP wrkWrdCnt = 1 to MyLine.Records()
           MyWord.SetValue(MyLine.GetLine(wrkWrdCnt))
           IF MyWord.StartsWith('ls', False)
              IF numeric(MyWord.Sub(3, MyWord.Length() - 2))
                 MyLine.DeleteLine(wrkWrdCnt)
              END
           END

           IF MyWord.StartsWith('f', False)
              IF NUMERIC(MyWord.Sub(2, 1))
                 MyFontChar.SetValue(MyWord.GetValue())
                 MyFontChar.Split(' ')
                 IF MyFontChar.Records() > 1
                    MyFontChar.SetLine(2, '\<39>95')
                    MyFontChar.Join('')
                    MyLine.SetLine(wrkWrdCnt, MyFontChar.GetValue())
                 END
              END
           END
        END

        MyLine.Join('\')
     ELSE
        wrkPosBgn = 1
        wrkPosEnd = 0

        LOOP !1
           wrkPosEnd = MyLine.Instring('\<39>B7', , wrkPosBgn)

           IF MyLine.Instring('\leveltext')
              BREAK
           END

           IF wrkPosEnd
              wrkPosBgn = wrkPosEnd + 1
              wrkDlmBgn = 1

              LOOP !2
                 wrkDlmPos = MyLine.Instring('{{', , wrkDlmBgn)
                 IF wrkDlmPos
                    LOOP !3
                       wrkDlmEnd = MyLine.MatchBrackets('{{', '}', wrkDlmPos)
                       IF wrkDlmEnd
                          wrkReplace &= NEW(STRING(MyLine.Length()))
                          wrkReplace  = MyLine.Slice(wrkDlmPos, wrkDlmEnd)

                          IF INSTRING('\<39>B7', wrkReplace, 1, 1)
                             IF    INSTRING('\pntext', wrkReplace, 1, 1)
                                   MyLine.Replace(CLIP(wrkReplace), '\cf0\bullet\tab\cf1 ')
                                   wrkDlmBgn = 1

                             ELSIF INSTRING('\pntxtb', wrkReplace, 1, 1)
                                   MyLine.Replace(CLIP(wrkReplace), '')
                                   wrkDlmBgn = 1
                             ELSE
                                   wrkDlmBgn = wrkDlmPos + 1
                             END
                          ELSE
                             wrkDlmBgn = wrkDlmEnd + 1 !Added KBE
                          END !IF INSTRING('\<39>B7', wrkReplace, 1, 1)

                          DISPOSE(wrkReplace)

                          BREAK
                       ELSE
                          BREAK
                       END !IF wrkDlmEnd
                    END !LOOP !3
                 ELSE
                    BREAK
                 END !IF wrkDlmPos
              END
           ELSE
              BREAK
           END !LOOP !2
        END !LOOP !1
     END !IF MyLine.Instring('\listtext')

     MyDoc.SetLine(wrkLineNo, MyLine.GetValue())

  END !LOOP wrkLineNo = 1 to MyDoc.Records()
!-
EXIT
!==============================================================================================

on 6/20/2017, Jorge A. Lavera supposed :

Hi, all.

I have a customer complaining that some reports using RTF are printing bullets and other
graphics as ? instead of the graphic. I verified and found the problem happens only in
Windows 10 (Works as expected in Windows 7).
Any idea on what to do?

Kind regards,
Jorge Lavera

--
Kevin Erskine
KErskine@Software-By-Ragazzi.com

